A Macro to Facilitate the Visualization of Individual Patient Efficacy and Safety During an Investigational (Phase III) Clinical Trial

Marjorie Bell, Applied Logic Associates, Inc., Houston, Texas
Marcella D. Stricklin, Applied Logic Associates, Inc., Houston, Texas

ABSTRACT

Intensive statistical analysis is usually performed on data collected in clinical trials to describe both the efficacy and safety of a new or investigational drug for the study population as a whole. However, the clinical investigators and FDA reviewers frequently have an interest in summarizing the course of the study for an individual patient including the patient's study medication dosage, plasma concentration levels, and efficacy as well as safety responses to the drug. A single page of customized graphs allow the investigator to view the individual patient's responses to the study medication with respect to the patient's dosing and toxicity information, all within the same time frame. A macro written for SAS/GRAPH® software referred to as RUNGPH accomplishes this goal. The macro uses the ANNOTATE, GPLOT and GREPLAY procedures in SAS/GRAPH. Examples of the use of the macro and the output produced are presented and discussed.

INTRODUCTION

In the pharmaceutical industry, the primary goal for conducting a clinical trial is to provide sufficient data to support both the efficacy and safety of a new or investigational drug. The pharmaceutical sponsor and clinical investigators collect data on numerous and various endpoints, and these response measures are usually statistically analyzed to describe both the efficacy and safety of the drug for the study population as a whole. Often, however, the investigators have an interest in summarizing the course of the study for an individual patient. The FDA also has an interest in examining the data on an individual patient basis since several divisions of the agency require case report tabulations (listings of all data for a patient) as part of an NDA (new drug application) submission. The patient's data may include study medication dosage including plasma concentration levels, and one or more efficacy responses as well as safety parameters such as vital signs, laboratory parameters and adverse event experiences. Graphics in general allow the investigator to visualize responses across time. A single page of customized graphics displaying the numerous study parameters collected for a given patient provides a detailed summary of the course of time and events for the patient. These customized graphs permit the investigator to view the individual patient's responses to the study medication with respect to the patient's dosing and toxicity information, all within the same time frame. Additionally, these individual patient graphs can be used as part of an NDA submission and could expedite the review process.

A SAS® macro, RUNGPH, is written to create a series of graphs specific to an individual patient which can be used to visualize the patient's efficacy and safety responses with respect to the study medication dosage throughout the study. The macro uses the ANNOTATE, GPLOT and GREPLAY procedures in SAS/GRAPH.

The single data set used in the macro consists of several data sets previously appended together from different study protocols. Also, four different data sets containing data on adverse events, percent change in seizure frequency, plasma concentration levels and study medication dosage were previously merged together by patient and date. The patients undergoing these study protocols began treatment in one of two double blind protocols and continued into one of two open label protocols. A global identifier was assigned to each patient to track the patient from the double blind protocol to the open label protocol.

A macro is utilized so that one patient graph is created per macro call. The macro extracts the patient's data from the data set using a WHERE statement. ANNOTATE data sets are created which contain the adverse event information and the title for the graph. The plots for percent change in seizure frequency, plasma concentration and study drug dosage are stored by using the GOUT option of the GPLOT procedure. These five graphs are plotted together on the same page using the GREPLAY procedure. The macro code and examples of its output are presented below.

PROGRAM EXAMPLE AND OUTPUT

An example demonstrating the use and output produced by the macro RUNGRH is presented below. Patient 04014 ingested 3600 mg of active study medication per day for nearly three years. The plasma concentration levels (µg/ml) for this patient tended to be somewhat unstable but showed a slight increase over time. The patient demonstrated a steady reduction in seizure frequency beginning about day 100 for 360 days. After receiving active treatment for more than 500 days, the patient revealed an increase in the percent change in seizure frequency which coincided with a decrease in plasma concentration levels. This decrease in plasma concentration may be related to the injury or rectal disorder the patient experienced after 500 days of treatment. After almost three years of treatment with active study drug, patient 04014 experienced greater than a 75% reduction in seizure frequency. Nevertheless, as the plasma concentration levels in the patient slightly increased, the seizure frequency decreased dramatically while the frequency and duration of adverse events increased across time.

CONCLUSION

Graphics, in general, are powerful tools for the visualization of responses across time. Clinical trial investigators and FDA reviewers frequently have an interest in summarizing the course of the study for an individual patient rather than the study population as a whole. Customized graphs generated by a SAS macro named RUNGPH allow the investigator to visualize the individual patient's responses to the study medication with respect to the patient's dosing and toxicity information, all within the same time frame on a single page. These individual patient graphs can also assist an FDA reviewer in determining the relationship between dosage and toxicity and thus, accelerating the NDA review process.

TRADEMARKS

REFERENCES

Mathieu, M., (1990), New Drug Development: A Regulatory Overview, Cambridge, MA: PAREXEL International Corporation, 132.

SAS/GRAPH® User's Guide, Release 6.03 Edition

CONTACT INFORMATION

Marjorie Bell
Applied Logic Associates
5615 Kirby Drive
Suite 710
Houston, TX 77005-2495

Phone: (713) 529-4747

```
************************************************************;
* Client:        SUGI 19                                   *;
*                                                          *;
* Program:       LTERM_G.SAS                               *;
*                                                          *;
* Output:        LTERM1.GPH - LTERM45.GPH                  *;
*                                                          *;
* Data Files:    LTERM.SSD                                 *;
*                                                          *;
* Description:   This macro creates long term efficacy and *;
*                safety graphs for patients who continued  *;
*                from the double blind protocols to the    *;
*                open label protocols.                     *;
*                                                          *;
*                Graph includes adverse events, percent    *;
*                change in seizure frequency, study drug   *;
*                plasma concentration and study drug       *;
*                dosage by study day.                      *;
*                                                          *;
*                                                          *;
* Programmer:    Marjorie Bell                             *;
*                Marcella D. Stricklin                     *;
*                                                          *;
* Date:          April 10, 1994                            *;
************************************************************;
LIBNAME SUGI 'P:\SUGI' ;

OPTIONS NODATE NONUMBER
        PAGESIZE = 57
        LINESIZE = 165 ;


%MACRO RUNGPH (PAT, FILE);

******************;
* PRINTER OPTIONS *;
******************;

GOPTIONS RESET=ALL DEVICE=HPLEGAL CTEXT=BLACK ROTATE=PORTRAIT
        GSFNAME=GRAFOUT GSFMODE=REPLACE NOPROMPT ;

SYMBOL1  C=BLACK V=DIAMOND H=0.25 CM I=J ;

*************************;
* SCREEN DISPLAY OPTIONS *;
*************************;

/**
GOPTIONS RESET=ALL DEV=WIN CTEXT=YELLOW FTEXT=DUPLEX ;

SYMBOL1  C=CYAN  V=DIAMOND H=0.5 CM  I=J ;
**/

AXIS2  LABEL=(A=90 R=0 H=.15 CM) LENGTH=17 PCT ;  * Y-AXIS ;


DATA T1 (KEEP=TREAT PATID ST_DAY GROUP ADVERSE ST_DAYCE
              ST_DAYON DOSEN SEIZ_C CONC) ;
   SET SUGI.LTERM ;
   WHERE PATID = "&PAT" ;


   ** ONLY CREATE GRAPHS FOR PATIENTS TREATED ** ;
   **             WITH STUDY DRUG             ** ;
   IF TREAT = 'STUDY DRUG' ;


   ** INTERESTED IN TREATMENT EMERGENT ADVERSE EVENTS ** ;
   IF ST_DAY GE 0 ;


   ** ASSUME DOSE OF 3600 MG FOR PROTOCOLS 366 & 368 ** ;
   IF (PROTOCOL = '366' OR PROTOCOL = '368' ) AND DOSE = .
     THEN DOSE = 3600 ;


   ** CONVERT CHARACTER DOSE VALUE TO NUMERIC ** ;
   DOSEN = DOSE / 1 ;


   ** CONVERT CHARACTER PLASMA CONCENTRATION TO NUMERIC ** ;
   IF PARENT NE '    BLQ' THEN CONC = PARENT / 1 ;


PROC SORT DATA = T1 ;
  BY TREAT PATID ;


** FIND MIN & MAX STUDY DAY.  USE MAX STUDY DAY WHEN **;
**          DATE CEASED OF AE IS ONGOING            **;

PROC MEANS DATA = T1 NOPRINT ;
  BY TREAT PATID ;
  VAR ST_DAY ;
OUTPUT OUT = M1 MIN = MIN_DAY
                MAX = MAX_DAY ;


DATA T1 ;
  MERGE T1
        M1 ;
  BY TREAT PATID ;

  IF ST_DAYCE = . THEN ST_DAYCE = MAX_DAY ;


DATA T2 ;
   SET T1 ;

  ** ONLY KEEP RECORDS WITH AES FOR GRAPHIC BARS ** ;
   IF ADVERSE NE ' ' ;

PROC SORT DATA = T2 ;
   BY TREAT PATID ADVERSE ST_DAYON ;


** CREATE AE GRAPHIC BARS ** ;
DATA ANNO ;
   SET T2 ;
   BY TREAT PATID ADVERSE ST_DAYON ;

   LENGTH COLOR FUNCTION STYLE $8 TEXT $21 ;
   RETAIN XSYS '2' YSYS '1' COLOR 'BLACK' SIZE 1 Y .5
          WHEN 'A' ;

   IF FIRST.ADVERSE THEN DO ;
    FUNCTION = 'MOVE' ; X = ST_DAYON ; Y = Y+5 ; STARTY=Y ;
     OUTPUT ;
    FUNCTION = 'BAR' ; X = ST_DAYCE ; Y = Y+1.5 ; ENDY = Y ;
     STYLE = 'X1' ; LINE = 0 ;
     OUTPUT ;
    FUNCTION = 'LABEL' ; X = (ST_DAYON+ST_DAYCE)/2; Y = Y ;
     POSITION = '2' ;  STYLE = 'SWISSL' ;
     TEXT = ADVERSE ; CHK1 = X + LENGTH(ADVERSE) + 1;
     CHK2 = X - LENGTH(ADVERSE) - 1 ;
     OUTPUT ;
   END ;

   ELSE DO ;
    FUNCTION = 'MOVE' ; X = ST_DAYON ; Y = STARTY ;
     OUTPUT ;
    FUNCTION = 'BAR' ; X = ST_DAYCE ; Y = ENDY ;
     STYLE = 'X1' ; LINE = 0 ;
     OUTPUT ;

    ** CHECK THAT AE NAMES WILL NOT OVERLAP ** ;
    IF (ST_DAYON+ST_DAYCE)/2 GT CHK1
      AND (ST_DAYON+ST_DAYCE)/2 LT CHK2 THEN DO ;
        FUNCTION = 'LABEL' ; X = (ST_DAYON+ST_DAYCE)/2 ;
         Y = Y ; POSITION = '2' ;  STYLE = 'SWISSL' ;
         TEXT = ADVERSE ;
         OUTPUT ;
    END ;
   END ;

RETAIN STARTY ENDY CHK1 CHK2 ;

PROC SORT DATA = ANNO ;
  BY TREAT PATID ST_DAYON ;
```

```
** CREATES TITLE ** ;
DATA ANNO2 ;
   LENGTH COLOR FUNCTION STYLE $8 TEXT $35 ;
   RETAIN XSYS YSYS '1' COLOR 'BLACK' SIZE 2 X 50
          POSITION '2'
   FUNCTION 'LABEL' STYLE 'SWISSL' ;

   TEXT="LONG TERM EFFICACY AND SAFETY GRAPH" ; Y=95 ;
    OUTPUT ;
   TEXT="PATIENT ID = &PAT" ; SIZE=1.5 ; Y=93 ;
    OUTPUT ;

PROC SORT DATA = T1 ;
   BY TREAT PATID ST_DAY ;


** OUTPUT DOSE VS STUOY OAY GRAPH ** ;
FILENAME  GRAFOUT  'P:\SUGI\TEMPG.OUT' ;
PROC GPLOT DATA = T1 GOUT = GRAPH&FILE ;
   PLOT DOSEN * ST_DAY  /
                          HAXIS = AXIS1
                          VAXIS = AXIS2 ;
SYMBOL1  C=BLACK  V=DIAMOND     H=0.25 CM  I=J ;
AXIS1 ORIGIN=(15 PCT,6.5 PCT) ;
   LABEL TREAT = 'TREATMENT'
         PATID = 'PATIENT ID'
         ST_DAY = 'STUDY DAY'
         DOSEN  = 'DOSAGE'
         CONC   = 'PLASMA CONC'
         SEIZ_C = '% CHANGE' ;
RUN ;
QUIT ;


** OUTPUT CONCENTRATION VS STUDY DAY GRAPH ** ;
FILENAME  GRAFOUT  'P:\SUGI\TEMPG.OUT' ;
PROC GPLOT DATA = T1 GOUT = GRAPH&FILE ;
   PLOT CONC  * ST_DAY  /
                          HAXIS = AXIS1
                          VAXIS = AXIS2 ;
   AXIS1 ORIGIN=(15 PCT,28 PCT) ;
   LABEL TREAT = 'TREATMENT'
         PATID = 'PATIENT ID'
         ST_DAY = 'STUDY DAY'
         DOSEN  = 'DOSAGE'
         CONC   = 'PLASMA CONC'
         SEIZ_C = '% CHANGE' ;
RUN ;
QUIT ;


** OUTPUT % CHANGE IN SEIZURE FREQ VS STUDY DAY GRAPH ** ;
FILENAME  GRAFOUT  'P:\SUGI\TEMPG.OUT' ;
PROC GPLOT DATA = T1 GOUT = GRAPH&FILE ;
   PLOT SEIZ_C * ST_DAY /
                          HAXIS = AXIS1
                          VAXIS = AXIS2 ;
   AXIS1 ORIGIN=(15 PCT,51 PCT) ;
   LABEL TREAT = 'TREATMENT'
         PATID = 'PATIENT ID'
         ST_DAY = 'STUDY DAY'
         DOSEN  = 'DOSAGE'
         CONC   = 'PLASMA CONC'
         SEIZ_C = '% CHANGE' ;
RUN ;
QUIT ;


** OUTPUT ADVERSE EVENTS ** ;
FILENAME  GRAFOUT  'P:\SUGI\TEMPG.OUT' ;
PROC GPLOT DATA = T1 ANNOTATE = ANNO GOUT = GRAPH&FILE ;
   PLOT DOSEN * ST_DAY  /
                          HAXIS = AXIS1
                          VAXIS = AXIS4 ;
AXIS1 ORIGIN=(15 PCT,72 PCT) ;

** ONLY WANT X-AXIS ** ;
AXIS4 VALUE=NONE LABEL=NONE LENGTH=23 PCT STYLE=0 MAJOR=NONE;
SYMBOL  C=BLACK  V=NONE  H=0.25 CM  I=NONE ;
   LABEL TREAT = 'TREATMENT'
         PATID = 'PATIENT ID'
         ST_DAY = 'STUDY DAY'
         DOSEN  = 'DOSAGE'
         CONC   = 'PLASMA CONC'
         SEIZ_C = '% CHANGE' ;
RUN ;
QUIT ;


** OUTPUT TITLE ** ;
PROC GANNO ANNOTATE = ANNO2 GOUT=GRAPH&FILE ;
RUN ;
QUIT ;


** REPLAYS 5 GRAPHS ON ONE PAGE ** ;
**    STORES IN GRAPHICS FILE   ** ;

FILENAME  GRAFOUT  "P:\SUGI\LTERM&FILE..GPH" ;
PROC GREPLAY IGOUT = GRAPH&FILE NOFS ;
  TC TEMPTEMP ;
  TDEF FULLPAGE
      1/LLX=0    LLY=0
        ULX=0    ULY=100
        URX=100  URY=100
        LRX=100  LRY=0   ;
  TEMPLATE FULLPAGE ;
  TREPLAY 1:1 1:2 1:3 1:4 1:5 ;
RUN ;
QUIT ;
%MEND RUNGPH ;

%RUNGPH(02006, 1);
%RUNGPH(03005, 2);
%RUNGPH(04014, 9);
```

LONG TERM EFFICACY AND SAFETY GRAPH

PATIENT ID = 04014